AD-A156 736

DTIC
JUL 1 1 1985
S D
G

DISTRIBUTION STATEMENT A

Approved for public release;
Distribution Unlimited

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

REPORT DOCUMENTATION PAGE		READ INSTRUCTIONS BEFORE COMPLETING FORM
1. REPORT NUMBER NRL Memorandun Report 5291	2. GOVT ACCESSION NO.	3. RECIPIENT'S CATALOG NUMBER
4. TITLE (and Subtitle) THE LIU, KATSAROS, AND BUSINGER (1979) BULK ATMOSPHERIC FLUX COMPUTATIONAL ITERATION PROGRAM IN FORTRAN AND BASIC		5. TYPE OF REPORT & PERIOD COVERED Final report on one phase of a continuing NRL problem. 6. PERFORMING ORG. REPORT NUMBER
7. AUTHOR(s) W. Timothy Liu* and Theodore V. Blanc		8. CONTRACT OR GRANT NUMBER(s)
9. PERFORMING ORGANIZATION NAME AND ADDRESS Naval Research Laboratory Code 4110 Washington, DC 20375		10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS 61153N; RR0330242; 41-1765-A-3
11. CONTROLLING OFFICE NAME AND ADDRESS Office of Naval Research NRL Non-Special Focus Program Washington, DC 20375		12. REPORT DATE May 8, 1984 13. NUMBER OF PAGES 20
14. MONITORING AGENCY NAME & ADDRESS(if different from Controlling Office)		15. SECURITY CLASS. (of this report) UNCLASSIFIED 15a. DECLASSIFICATION/DOWNGRADING SCHEDULE

16. DISTRIBUTION STATEMENT (of this Report)

Approved for public release; distribution unlimited.

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

*Earth and Space Sciences Division, Jet Propulsion Laboratory, Pasadena, CA 91109

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

Bulk derived surface flux method
Air-sea interaction

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

The computer program described by Liu, Katsaros, and Businger (1979) for calculating the bulk-derived atmospheric fluxes, stability, and roughness is presented in both FORTRAN and BASIC versions. The results of 12 test calculations are presented to ensure that the program has been properly encoded.

DD FORM 1 JAN 73 1473 EDITION OF 1 NOV 65 IS OBSOLETE
S/N 0102-014-6601

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

CONTENTS

Accession For
NTIS GRA&I ☒
DTIC TAB ☐
Unannounced ☐
Justification

By
Distribution/
Availability Codes

Dist	Avail and/or Special
A/1	

DTIC COPY

THE LIU, KATSAROS, AND BUSINGER (1979) BULK ATMOSPHERIC FLUX COMPUTATIONAL ITERATION PROGRAM IN FORTRAN AND BASIC

ABSTRACT

The computer program described by Liu, Katsaros, and Businger (1979) for calculating the bulk-derived atmospheric fluxes, stability, and roughness is presented in both FORTRAN and BASIC versions. The results of 12 test calculations are presented to ensure that the program has been properly encoded.

INTRODUCTION

Liu, Katsaros, and Businger (1979) present a model developed for the marine atmospheric surface layer which takes into consideration the interfacial sublayers on both sides of the air-sea interface where molecular constraints on transports are important. Flux-profile relations which are based on the postulation of intermittent renewal of the surface fluid are matched to the logarithmic profiles and compared with both field and laboratory measurements. These relations enable numerical determination of air-sea exchanges of momentum, heat, and water vapor (or bulk transfer coefficients) employing the bulk parameters of mean wind speed, temperature, and humidity at a certain height in the atmospheric surface layer and the water temperature.

With increasing wind speed, the flow goes from smooth to rough and the bulk transfer coefficient for momentum also increases. The increase in roughness is associated with increasing wave height, which in the present model results in sheltering at the wave troughs. Due to the decrease in turbulent transport, the transfer coefficients of heat and water vapor decrease slightly with wind speed after the wind speed exceeds a certain value. The bulk transfer coefficients are also found to decrease with increasing stability. By including the effects of stability and interface conditions in bulk parameterization, the model provides a way to account for physical conditions which are known to affect air-sea exchanges.

A computer program utilizing computational iteration is required in order to solve three simultaneous equations of three unknowns, which are formed from five equations containing twelve dependent variables. Although the original program was written in FORTRAN, the recent proliferation of compact desk-top computers usable in the field has warranted its translation into BASIC.

The program requires seven input parameters:

1. Wind speed (u) in m/s;
2. Air temperature (T) in °C;
3. Specific humidity (q) in kg/kg;
4. Water temperature at the surface (T_S) in °C;
5. Altitude of the wind-speed measurement (z_u) in m;
6. Altitude of the air-temperature measurement (z_t) in m; and
7. Altitude of the humidity measurement (z_q) in m.

Manuscript Approved February 24, 1984

If the humidity is to be inputted in terms of the wet bulb temperature, dew point temperature, or relative humidity, consult the Appendix at the end of this report.

The results are outputted in the form of eight parameters:

1. The friction velocity (u_*) in m/s;
2. The scaling potential temperature (θ_*) in °C;
3. The scaling specific humidity (q_*) in kg/kg;
4. The roughness length (z_0) in m;
5. The Monin-Obukhov stability (z/L);
6. The wind-speed-roughness Reynold's number (R_r);
7. The temperature-roughness Reynold's number (R_t); and
8. The humidity-roughness Reynold's number (R_q).

From the outputted information it is possible to compute the momentum flux (M) in N/m², the sensible heat flux (H_S) in W/m², and the humidity flux (E) in kg/(s m²) by

$$M = -\rho u_*^2,$$

$$H_S = -\rho\, C_p\, \theta_*\, u_*,$$

and

$$E = -\rho\, q_*\, u_*,$$

where the density of moist air (ρ) in kg/m³ and the specific heat of moist air at constant pressure (C_p) in J/(kg/K) can be determined by

$$\rho = \frac{(3.4838 \times 10^{-3})P}{T_v + 273.16}$$

and

$$C_p = 1.004[1 + 0.9(q)] \times 10^3,$$

where P is the barometeric pressure in pascals (if not recorded, assumed to be 1.01325×10^5 Pascals) and T_v is the virtual potential temperature in °C such that

$$T_v = \{(T + 273.16) \times [1 + (0.608q)]\} - 273.16.$$

By convention, a positive sign is used to indicate an upward flux and a negative sign a downward flux. Because momentum flux is almost always downward (except in a few cases when the water velocity exceeds the wind), the negative momentum flux is traditionally defined as the stress (τ),

$$\tau \equiv -M.$$

Additionally, it is frequently convenient from a thermodynamic perspective to represent the humidity flux in terms of the latent heat flux (H_L) in W/m² such that

$$H_L \equiv E\, L_v$$

where L_v is the latent heat of vaporization in J/kg,

$$L_v = 4.1868(597.31 - 0.56525\, T) \times 10^3.$$

THE FORTRAN PROGRAM

```
      PROGRAM LKB79F
      COMMON/PIN/U,T,Q,TS,ZU,ZT,ZQ,ID
      COMMON/POUT/USR,TSR,QSR,ZO,ZL,RR,RT,RQ
C
C INPUT DATA FROM TTY
C
 1000 CONTINUE
      USR=0.0
      TSR=0.0
      QSR=0.0
      ZO=0.0
      ZL=0.0
      RR=0.0
      RT=0.0
      RQ=0.0
      CD=0.0
      WRITE (5,1)
 1    FORMAT(' INPUT VALUES ARE:')
      WRITE (5,2)
 2    FORMAT(' U=  ', $)
      READ(5,900) U
      WRITE(5,3)
 3    FORMAT(' T= ', $)
      READ(5,900)T
      WRITE(5,4)
 4    FORMAT(' Q= ',$)
      READ(5,900)Q
      WRITE(5,5)
 5    FORMAT(' TS= ', $)
      READ(5,900)TS
      WRITE(5,6)
 6    FORMAT(' ZU= ', $)
      READ(5,900)ZU
      WRITE(5,7)
 7    FORMAT(' ZT= ', $ )
      READ(5,900)ZT
      WRITE(5,8)
 8    FORMAT(' ZQ= ',$)
      READ(5,900)ZQ
      CALL ASL(IER)
      WRITE(5,20)USR,TSR,QSR,ZO,ZL,RR,RT,RQ,IER
 20   FORMAT(' OUTPUT VALUES ARE:',/,' USR= ',G13.6,/,
     A                               ' TSR=',G13.6,/,
     B                               ' QSR=',G13.6,/,
     C                               ' ZO =',G13.6,/,
     D                               ' ZL =',G13.6,/,
     E                                 ' RR =',G13.6,/,
     F                                 ' RT =',G13.6,/,
     G                                 ' RQ =',G13.6,/,' IER=',G13.6)
 900  FORMAT(G13.6)
      WRITE(5,910)
  910 FORMAT(' LAST CASE? 0=YES, 1=NO ',$)
      READ(5,920)IEND
  920 FORMAT(I1)
      IF (IEND.EQ.1)GO TO 1000
      END
      SUBROUTINE ASL(IER)
C
C TO EVALUATE SURFACE FLUXES, SURFACE ROUGHNESS, AND STABILITY OF
C THE ATMOSPHERIC SURFACE LAYER FROM BULK PARAMETERS ACCORDING TO
C LIU EL AL. (79) JAS 36 1722-1735
C WRITTEN BY TIM LIU ON 5/8/79
C
C INPUT:
C U WIND SPEED IN M/S
```

```
C T TEMPERATURE IN DEG C
C Q SPECIFIC HUMIDITY IN KG/KG
C TS SURFACE TEMPERATURE IN DEG C
C ZU HEIGHT OF WIND SENSOR IN METERS
C ZT HEIGHT OF TEMPERATURE SENSOR
C ZQ HEIGHT OF HUMIDITY SENSOR
C ID SEE SUBROUTINE DRAG FOR DETAIL DEFINITION,ID=1 (KONDO),
C    ID+2 (SMITH),ID=3 (LARGE)
C OUTPUT:
C USR,TSR,QSR SCALING QUANTITIES FOR U,T,Q
C ZO,ZL ROUGHNESS AND STABILITY PARAMETERS
C RR,RT,RQ ROUGNESS REYNOLD NUMBERS FOR U,T,Q
C
C IER=1 FAIL TO CONVERGE
C IER=2 LBK ERROR
C
        COMMON/PIN/U,T,Q,TS,ZU,ZT,ZQ,ID
        COMMON/POUT/USR,TSR,QSR,ZO,ZL,RR,RT,RQ
        IER=0
        RI=9.81*ZU*(T-TS)/((273.15+T)*U**2)
        IF(RI.GT.0.25)IER=-1
        VISA=.15E-4
        ZL=0.
        ZO=.0005
        US=0.
        CALL HUMLOW(TS,TS,QS)
        DU=U-US
        DT=T-TS
        DQ=Q-QS
        USR=.04*DU
        N3=0
 30     CONTINUE
        N1=0
 10     CONTINUE
        U10=USR*ALOG(10./ZO)/.4
        TYPE 8400,ID,U10,CD
 8400   FORMAT (' DRAG ' ,3G13.6)
        CALL DRAG(ID,U10,CD)
        TYPE 8400,ID,U10,CD
        C=1./SQRT(CD)
        ZON=10./EXP(.4*C)
        TEST1=ABS((ZON-ZO)/(ZO+1.E-8))
        TYPE 8810,TEST1,ZON,ZO,C,CD,N1
 8810   FORMAT (' TEST1,ZON,ZO,C,CD,N1',6G13.6)
        IF(TEST1.LT.0.01)GO TO 19
        N1=N1+1
        IF(N1.GT.50)GO TO 95
        ZO=ZON
        GO TO 10
 19     CONTINUE
        PUZ=PSI(1,ZL)
        ZTL=ZL*ZT/ZU
        ZQL=ZL*ZQ/ZU
        PTZ=PSI(2,ZTL)
        PQZ=PSI(2,ZQL)
        USR=DU*0.4/(ALOG(ZU/ZO)-PUZ)
        RR=ZO*USR/VISA
        ZTSR=ZT*USR/VISA
        ZQSR=ZQ*USR/VISA
        CALL LKB(RR,RT,1)
        IF(RT .NE. -999.)GO TO 21
        IER=2
        WRITE(5,2)RR
        RETURN
 21     CALL LKB(RR,RQ,2)
        IF(RT .NE. -999.)GO TO 22
```

```
      IER=2
      WRITE(5,2)RR
      RETURN
22    S=2.2*(ALOG(ZTSR/RT)-PTZ)
      D=2.2*(ALOG(ZQSR/RQ)-PQZ)
      TSR=DT/S
      QSR=DQ/D
      CALL ZETA(T,Q,USR,TSR,QSR,ZU,ZLN)
      TEST3=ABS((ZL-ZLN)/(ZL+1.E-8))
      IF(TEST3.LT.0.01)GO TO 39
      N3=N3+1
      IF(N3.GT.50)GO TO 95
      ZL=ZLN
      GO TO 30
39    CONTINUE
      GO TO 99
95    IER=1
      WRITE(5,1)N1,N3
1     FORMAT(1X,21HASL FAILS TO CONVERGE,3I5)
2     FORMAT(1X,21HLKB FAILS BECAUSE RR=,E12.4)
99    RETURN
      END
      SUBROUTINE DRAG(ID,U,CD)
C
C TO DETERMINE NEUTRAL DRAG COEFFICIENT CD FROM WIND SPEED
C AT 10 M U IN M/S
C ID=1 KONDO(1975) BLM 9 91-112
C ID=2 SMITH(1980) JPO 10 709-726
C ID=3 LARGE & POND (1981) JPO 11 324-336
C RANGE OF U SPECIFIED ARE: KONDO(.3,50),SMITH(6,22),L&P(4,25)
C WRITTEN BY TIM LIU FOR VAX ON 2/10/82
C
      DIMENSION RAN(5),A(5),B(5),P(5)
      DATA RAN/2.2,5.,8.,25.,50./
      DATA A/0.,0.771,0.867,1.2,0./
      DATA B/1.08,0.0858,.0667,0.025,0.073/
      DATA P/-0.15,1.,1.,1.,1./
      K=ID-2
      IF(K)100,200,300
100   IF(U.GT.50.)GO TO 131
      IF(U.LT..3)GO TO 130
      I=1
110   IF(U.LE.RAN(I))GO TO 120
      I=I+1
      GO TO 110
120   CD=(A(I)+B(I)*U**P(I))/1000.
      GO TO 99
130   CD=1.5E-03
      GO TO 99
131   CD=3.7E-03
      GO TO 99
200   CD=(0.61+0.063*U)/1000.
      GO TO 99
300   IF(U.LT.11.)GO TO 301
      CD=(0.49+0.065*U)/1000.
      GO TO 99
301   CD=1.2E-3
99    RETURN
      END
      SUBROUTINE HUMLOW(T,TW,Q)
C
C TO EVALUATE SPECIFIC HUMIDITY Q FROM DRY AND WET BULB TEMP
C T AND TW ACCORDING TO LOWE(77) JAM 16 100-103
C WRITTEN BY TIM LIU ON 5/3/79, REVISED FOR VAX ON 2/10/82
C
      DIMENSION A(6)
```

```
      DATA A/4.436519E-1,1.428946E-2,2.650649E-4,3.031240E-6,
     $ 2.034081E-8,6.136821E-11/
      P=1013.25
      X=0.
      DO 100 I=1,6
      J=7-I
      X=(X+A(J))*TW
100   CONTINUE
      ES=6.107800+X
      Q=0.622*ES/(P-ES)-4.045E-04*(T-TW)
      RETURN
      END
      SUBROUTINE LKB(RR,RT,IFLAG)
C
C TO DETERMINE THE LOWER BOUNDRY VALUE RT OF THE LOGARITHMIC
C PROFILES OF TEMPERATURE (IFLAG=1) OR HUMIDITY (IFLAG=2)
C IN THE ATMOSPHERE FROM ROUGHNESS REYNOLD NUMBER RR BETWEEN
C 0 AND 1000. OUT OF RANGE RR INDICATED BY RT=-999.
C BASED ON LIU ET AL.. (1979) JAS 36 1722-1723
C WRITTEN BY TIM LIU ON 1/22/78, REVISED FOR VAX ON 2/10/82
C
      DIMENSION A(8,2),B(8,2),RAN(8)
      DATA A/0.177,1.376,1.026,1.625,4.661,34.904,1667.19,5.88E5,
     $ 0.292,1.808,1.393,1.956,4.994,30.709,1448.68,2.98E5/
      DATA B/0.,0.929,-0.599,-1.018,-1.475,-2.067,-2.907,-3.935,
     $ 0.,0.826,-0.528,-0.870,-1.297,-1.845,-2.682,-3.616/
      DATA RAN/0.11,0.825,3.0,10.0,30.0,100.,300.,1000./
      I=1
      IF (RR.LE.0..OR.RR.GE.1000.) GO TO 90
10    CONTINUE
      IF (RR.LE.RAN(I)) GO TO 20
      I=I+1
      GO TO 10
20    RT=A(I,IFLAG)*RR**B(I,IFLAG)
      GO TO 99
90    RT=-999.
99    RETURN
      END
      FUNCTION PSI(ID,ZL)
C TO EVALUATE THE STABILITY FUNCTION PSI FOR WIND SPEED (IFLAG=1)
C OR FOR TEMPERATURE AND HUMIDITY PROFILES FROM STABILITY PARAMETER ZL
C SEE LIU ET AL. (1979) JAS 36 1722-1723 FOR DETAILS
C WRITTEN BY TIM LIU ON 9/12/71, REVISED FOR VAX ON 2/10/82
C
      IF(ZL)10,20,30
10    CHI=(1.-16.*ZL)**0.25
      IF(ID.EQ.1)GO TO 11
      PSI=2.*ALOG((1.+CHI*CHI)/2.)
      GO TO 99
11    PSI=2.*ALOG((1.+CHI)/2.)+ALOG((1.+CHI*CHI)/2.)-2.*ATAN(CHI)
     &  +2.*ATAN(1.)
      GO TO 99
20    PSI=0.
      GO TO 99
30    PSI=-6.*ALOG(1.+ZL)
99    RETURN
      END
      SUBROUTINE ZETA(T,Q,USR,TSR,QSR,Z,ZL)
C
C TO EVALUATE OBUKHOVS STABILITY PARAMETER Z/L FROM AVERAGE
C TEMP T IN DEG C, AVERAGE HUMIDITY Q IN GM/GM, HEIGHT Z IN M,
C AND FRICTIONAL VEL,TEMP., HUM. INMKS UNITS
C SEE LIU ET AL. (1979) JAS 36 1722-1723 FOR DETAILS
C WRITTEN BY TIM LIU ON 10/1/77, REVISED FOR VAX ON 2/10/82
C
      VON=0.4
```

```
      G=9.81
      TA=273.16+T
      TV=TA*(1.+0.61*Q)
      TVSR=TSR*(1.+0.61*Q)+0.61*TA*QSR
      IF(TVSR.EQ.0.)GO TO 10
      OB=TV*USR*USR/(G*VON*TVSR)
      ZL=Z/OB
      GO TO 99
10    ZL=0.
99    RETURN
      END
```

THE BASIC PROGRAM

```
10    REM
20    REM    "LKB79B" PROGRAM    12 OCT 83    T.V. BLANC    NRL 4110    DISK #6
30    REM
40    REM    THE BULK FLUX COMPUTATIONAL ITERATION PROGRAM OF LIU, KATSAROS, &
50    REM    BUSINGER (1979) TRANSLATED FROM FORTRAN INTO ELEMENTRY BASIC FOR USE
60    REM    ON A HEWLETT-PACKARD MODEL 9845.  THIS PROGRAM WAS ADAPTED FROM AN
70    REM    ORGINAL TRANSLATION OF A T WILSON NRL 2820 FOR USE ON A DIGITAL
80    REM    EQUIPMENT CORPORATION MODEL DEC-10 COMPUTER.
90    REM
100   REM    INPUTS  U= Wind Speed (m/s), T= Air Temp (C), Q= Spec.Humid.(Kg/Kg),
110   REM            T1=Tw= Water Temp (C), Z1=Zu= U Altitude(m), Z2=Zt= T Alt.(m)
120   REM            Z3=Zq= Q Alt (m)
130   REM
140   REM    NOTES   U SHOULD NOT = 0
150   REM
160   REM    OUTPUTS. U1=Ustr= U* (m/s), T2=Tstr= T* (C), Q1=Qstr= Q* (Kg/Kg),
170   REM             Z4=Zo= Rough  Length (m), Z6=Z/L= Monin-Obukhov Stability,
180   REM             R1=Ru= U Rough  Reynolds No , R2=Rt= T Rough  Reynolds No ,
190   REM             R3=Rq= Q Rough  Reynolds No
200   REM
210   DIM R5(5),A2(5),B2(5),P(5)
220    DATA 2 2,5 ,8 ,25 ,50
230    DATA 0 ,0 771,0 867,1 2,0
240    DATA 1 08,0 0858,0 0667,0 025,0 073
250    DATA -0 15,1 ,1 ,1 ,1
260   DIM A1(6)
270    DATA 4 436519E-1,1 428946E-2,2 650649E-4
280    DATA 3 031240E-6,2 034081E-8,6 136821E-11
290   DIM A(8,2),B(8,2),R6(8)
300    DATA 0 177,1 376,1 026,1 625,4 661,34 904,1667 19,5.88E5
310    DATA 0 292,1 808,1 393,1 956,4 994,30 709,1448.68,2 98E5
320    DATA 0 ,0 929,-0 599,-1 018,-1 475,-2.067,-2.907,-3 935
330    DATA 0 ,0 826,-0 528,-0.870,-1 297,1.845,-2.682,-3 616
340    DATA 0 11,0 825,3 0,10 0,30 0,100 0,300 ,1000 0
350    RESTORE 220
360   REM
370   FOR I=1 TO 5
380   READ R5(I)
390   NEXT I
400   FOR I=1 TO 5
410   READ A2(I)
420   NEXT I
430   FOR I=1 TO 5
440   READ B2(I)
450   NEXT I
460   FOR I=1 TO 5
470   READ P(I)
480   NEXT I
490   FOR I=1 TO 6
500   READ A1(I)
510   NEXT I
520   FOR J=1 TO 2
530   FOR I=1 TO 8
540   READ A(I,J)
550   NEXT I
560   NEXT J
570   FOR J=1 TO 2
580   FOR I=1 TO 8
590   READ B(I,J)
600   NEXT I
610   NEXT J
620   FOR I=1 TO 8
630   READ R6(I)
640   NEXT I
650   REM
660   PRINT "    "
```

```
670  PRINT ".............................................................................................."
680  PRINT "INPUT DATA: U(m/s),T(C),Q(kg/kg),Tw(C),Zu(m),Zt(m),Zq(m)"
690  LET U1=0
700  LET T2=0
710  LET Q1=0
720  LET Z4=0
730  LET Z6=0
740  LET R1=0
750  LET R2=0
760  LET R3=0
770  INPUT U,T,Q,T1,Z1,Z2,Z3
780  PRINT U,T,Q,T1,Z1,Z2,Z3
790  REM ................................CALL SUROUTINE ASL(IER)
800  LET A9=I9
810  GOSUB 990
820  LET I9=A9
830  REM ...............................
840  PRINT "OUTPUT VALUES ARE: "
850  PRINT "U* (m/s) =",U1
860  PRINT "T* (C)   =",T2
870  PRINT "Q*(kg/kg)=",Q1
880  PRINT "Zo (m)   =",Z4
890  PRINT "Z/L      =",Z6
900  PRINT "Ru       =",R1
910  PRINT "Rt       =",R2
920  PRINT "Rq       =",R3
930  PRINT "    "
940  REM ........................................................................
950  PRINT "LAST CASE? O=YES, 1=NO ";
960  INPUT I8
970  IF I8=1 THEN GOTO 660
980  GOTO 2670
990  REM ...............................SUBROUTINE ASL(IER)
1000 LET A9=0
1010 LET R4=9.81*Z1*(T-T1)/((273.15+T)*U^2)
1020 IF R4>.25 THEN GOTO 1040
1030 GOTO 1050
1040 LET A9=-1
1050 LET V1=1.5E-5
1060 LET Z5=0
1070 LET Z4=.0005
1080 LET U2=0
1090 LET H1=T1
1100 LET H2=T1
1110 LET H3=Q1
1120 REM ........................CALL HUMLOW(Tw,Tw,QS)
1130 GOSUB 2290
1140 LET T1=H1
1150 LET T1=H2
1160 LET Q1=H3
1170 LET DO=U-U2
1180 LET W2=T-T1
1190 LET W3=Q-Q1
1200 LET U1=.04*DO
1210 LET N3=0
1220 REM ..........CONTINUE
1230 LET N1=0
1240 LET UO=U1*LOG(10/Z4)/.4
1250 LET D1=D1
1260 LET D2=UO
1270 LET D3=C1
1280 GOSUB 2050
1290 REM ........................CALL DRAG(ID,U10,CD)
1300 LET D1=D1
1310 LET UO=D2
1320 LET C1=D3
```

```
1330 LET C=1/SQR(C1)
1340 LET Z5=10/EXP(.4*C)
1350 LET T3=ABS((Z5-Z4)/(Z4+1E-8))
1360 IF T3<.01 THEN GOTO 1410
1370 LET N1=N1+1
1380 IF N1>50 THEN GOTO 2020
1390 LET Z4=Z5
1400 GOTO 1240
1410 REM ............ CONTINUE
1420 LET P1=FNP(1,Z6)
1430 LET Z7=Z6*Z2/Z1
1440 LET Z8=Z6*Z3/Z1
1450 LET P2=FNP(2,Z7)
1460 LET P3=FNP(2,Z8)
1470 U1=DO*.4/(LOG(Z1/Z4)-P1)
1480 LET R1=Z4*U1/V1
1490 LET Z9=Z2*U1/V1
1500 LET Z0=Z3*U1/V1
1510 LET L1=R1
1520 LET L2=R2
1530 LET L3=1
1540 GOSUB 2410
1550 REM ..................,.CALL LKB(Ru,Rt,1)
1560 LET R1=L1
1570 LET R2=L2
1580 LET L3=L3
1590 IF R2<>-999 THEN GOTO 1630
1600 LET A9=2
1610 PRINT "LBK FAILS BECAUSE Ru=",R1
1620 RETURN
1630 REM ........SET VALUE OF 2 TO A VARIABLE
1640 LET L1=R1
1650 LET L2=R3
1660 LET L3=2
1670 GOSUB 2410
1680 LET R1=L1
1690 LET R3=L2
1700 LET L3=L3
1710 IF R2<>-.999 THEN GOTO 1740
1720 LET A9=2
1730 GOTO 1610
1740 LET S=2.2*(LOG(Z9/R2)-P2)
1750 LET D=2.2*(LOG(Z0/R3)-P3)
1760 LET T2=W2/S
1770 LET Q1=W3/D
1780 REM ............................CALL ZETA(T,Q,Ustr,Tstr,Qstr,ZU,ZLN)
1790 LET F1=T
1800 LET F2=Q
1810 LET F3=U1
1820 LET F4=T2
1830 LET F5=Q1
1840 LET F6=Z1
1850 LET F7=Y1
1860 GOSUB 2550
1870 LET T=F1
1880 LET Q=F2
1890 LET U1=F3
1900 LET T2=F4
1910 LET Q1=F5
1920 LET Z1=F6
1930 LET Y1=F7
1940 LET T4=ABS((Z6-Y1)/(Z6+1E-8))
1950 IF T4<.01 THEN GOTO 2000
1960 LET N3=N3+1
1970 IF N3>50 THEN GOTO 2020
1980 LET Z6=Y1
```

```
1990 GOTO 1220
2000 REM ..........CONTINUE
2010 GOTO 2040
2020 LET A9=1
2030 PRINT "ASL FAILS TO CONVERGE",N1,N3
2040 RETURN
2050 REM ..........................SUBROUTINE DRAG(D1,D2,D3)
2060 REM                           R5=RAN,A2=A,B2=B IN DRAG
2070 LET K=D1-2
2080 IF K=0 THEN GOTO 2220
2090 IF K>0 THEN GOTO 2240
2100 IF D2>50 THEN GOTO 2200
2110 IF D2<.3 THEN GOTO 2180
2120 LET I=1
2130 IF D2<=R5(I) THEN GOTO 2160
2140 LET I=I+1
2150 GOTO 2130
2160 LET D3=(A2(I)+B2(I)*D2^P(I))/1000
2170 GOTO 2280
2180 LET D3=1.5E-3
2190 GOTO 2280
2200 LET D3=3.7E-3
2210 GOTO 2280
2220 LET D3=(.61+.063*D2)/1000
2230 GOTO 2280
2240 IF D2<11 THEN GOTO 2270
2250 LET D3=(.49+.065*D2)/1000
2260 GOTO 2280
2270 LET D3=1.2E-3
2280 RETURN
2290 REM ...........................SUBROUTINE HUMLOW(T,TW,Q)
2300 REM                            SUBROUTINE HUMLOW(H1,H2,H3)
2310 REM                            A1=A IN HUMLOW
2320 LET P=1013.25
2330 LET X=0
2340 FOR I=1 TO 6
2350 LET J=7-I
2360 LET X=(X+A1(J))*H2
2370 NEXT I
2380 LET E1=6.107800+X
2390 LET H3=.622*E1/(P-E1)-4.045E-4*(H1-H2)
2400 RETURN
2410 REM ............................SUBROUTINE LKB(Ru,Rt,IFLAG)
2420 REM                             SUBROUTINE LKB(L1,L2,L3)
2430 REM                             R6=RAN IN LKB
2440 LET I=1
2450 IF L1<=0 THEN GOTO 2530
2460 IF L1>=1000 THEN GOTO 2530
2470 REM .......CONTINUE
2480 IF L1<=R6(I) THEN GOTO 2510
2490 LET I=I+1
2500 GOTO 2470
2510 LET L2=A(I,L3)*R1^B(I,L3)
2520 GOTO 2540
2530 LET L2=-999
2540 RETURN
2550 REM .............................SUBROUTINE ZETA (T,Q,Ustr,Tstr,Qstr,Z,Z/L)
2560 LET V=.4
2570 LET G=9.81
2580 LET T5=273.16+F1
2590 LET T6=T5*(1+.61*F2)
2600 LET T7=F4*(1+.61*F2)+.61*T5*F5
2610 IF T7=0 THEN GOTO 2650
2620 LET O1=T6*F3*F3/(G*V*T7)
2630 LET F7=F6/O1
2640 GOTO 2660
```

```
2650 LET F7=0
2660 RETURN
2670 END
2680 REM ..........................................................................
2690 REM ........................... FUNCTION PSI(ID,Z/L)
2700 DEF FNP(I7,X5)
2710 REM ..........TEST Z/L FOR -,0,+
2720 IF X5=0 THEN GOTO 2800
2730 IF X5>0 THEN GOTO 2820
2740 LET C2=(1-16*X5)^.25
2750 IF I7=1 THEN GOTO 2780
2760 RETURN 2*LOG((1+C2*C2)/2)
2770 GOTO 2830
2780 RETURN 2*LOG((1+C2)/2)+LOG((1+C2*C2)/2)-2*ATN(C2)+2*ATN(1)
2790 GOTO 2830
2800 RETURN 0
2810 GOTO 2830
2820 RETURN -6*LOG(1+X5)
2830 FNEND
```

TEST CALCULATION RESULTS

To ensure that the programs have been properly copied by the user, the results of 12 test calculations are presented. The outputted results should agree with those indicated below to within ±2 of the fourth significant figure.

Test	Input	Output
1	$u = 9$	$u_* = 0.3490$
	$T = 12$	$\theta_* = -6.922 \times 10^{-2}$
	$q = 0.007$	$q_* = -1.062 \times 10^{-4}$
	$T_s = 14$	$z_0 = 2.501 \times 10^{-4}$
	$z_u = 10$	$z/L = -9.937 \times 10^{-2}$
	$z_t = 10$	$R_r = 5.820$
	$z_q = 10$	$R_t = 0.2705$
		$R_q = 0.4226$

Test	Input	Output
2	$u = 9$	$u_* = 0.3262$
	$T = 14$	$\theta_* = 6.536 \times 10^{-2}$
	$q = 0.007$	$q_* = -5.817 \times 10^{-5}$
	$T_s = 12$	$z_0 = 2.429 \times 10^{-4}$
	$z_u = 10$	$z/L = 7.087 \times 10^{-2}$
	$z_t = 10$	$R_r = 5.282$
	$z_q = 10$	$R_t = 0.2986$
		$R_q = 0.4598$

Test	Input	Output
3	$u = 9$	$u_* = 0.3514$
	$T = 12$	$\theta_* = -6.981 \times 10^{-2}$
	$q = 0.002$	$q_* = -2.879 \times 10^{-4}$
	$T_s = 14$	$z_0 = 2.501 \times 10^{-4}$
	$z_u = 10$	$z/L = -0.1347$
	$z_t = 10$	$R_r = 5.859$
	$z_q = 10$	$R_t = 0.2686$
		$R_q = 0.4201$

Test	Input	Output
4	$u = 9$	$u_* = 0.3334$
	$T = 14$	$\theta_* = 6.614 \times 10^{-2}$
	$q = 0.002$	$q_* = -2.297 \times 10^{-4}$
	$T_s = 12$	$z_0 = 2.466 \times 10^{-4}$
	$z_u = 10$	$z/L = 3.175 \times 10^{-2}$
	$z_t = 10$	$R_r = 5.481$
	$z_q = 10$	$R_t = 0.2875$
		$R_q = 0.4452$

Test	Input	Output
5	$u = 2$	$u_* = 7.230 \times 10^{-2}$
	$T = 12$	$\theta_* = -9.516 \times 10^{-2}$
	$q = 0.007$	$q_* = -1.486 \times 10^{-4}$
	$T_s = 14$	$z_0 = 2.653 \times 10^{-5}$
	$z_n = 10$	$z/L = -3.175$
	$z_t = 10$	$R_r = 0.1279$
	$z_q = 10$	$R_t = 0.2036$
		$R_q = 0.3306$

Test	Input	Output
6	$u = 2$	$u_* = 3.731 \times 10^{-2}$
	$T = 14$	$\theta_* = 4.336 \times 10^{-2}$
	$q = 0.007$	$q_* = -3.831 \times 10^{-5}$
	$T_s = 12$	$z_0 = 4.498 \times 10^{-5}$
	$z_u = 10$	$z/L = 3.578$
	$z_t = 10$	$R_r = 0.1119$
	$z_q = 10$	$R_t = 0.1799$
		$R_q = 0.2962$

Test	Input	Output
7	$u = 18$	$u_* = 0.7361$
	$T = 12$	$\theta_* = -5.599 \times 10^{-2}$
	$q = 0.007$	$q_* = -8.643 \times 10^{-5}$
	$T_s = 14$	$z_0 = 5.289 \times 10^{-4}$
	$z_u = 10$	$z/L = -1.802 \times 10^{-2}$
	$z_t = 10$	$R_r = 25.96$
	$z_q = 10$	$R_t = 3.824 \times 10^{-2}$
		$R_q = 7.315 \times 10^{-2}$

Test	Input	Output
8	$u = 18$	$u_* = 0.7259$
	$T = 14$	$\theta_* = 5.542 \times 10^{-2}$
	$q = 0.007$	$q_* = -4.975 \times 10^{-5}$
	$T_s = 12$	$z_0 = 5.289 \times 10^{-4}$
	$z_u = 10$	$z/L = 1.203 \times 10^{-2}$
	$z_t = 10$	$R_r = 25.59$
	$z_q = 10$	$R_t = 3.904 \times 10^{-2}$
		$R_q = 7.449 \times 10^{-2}$

Test	Input	Output
9	$u = 9$	$u_* = 0.3250$
	$T = 12$	$\theta_* = -7.050 \times 10^{-2}$
	$q = 0.007$	$q_* = -1.081 \times 10^{-4}$
	$T_s = 14$	$z_0 = 2.421 \times 10^{-4}$
	$z_u = 30$	$z/L = -0.3487$
	$z_t = 10$	$R_r = 5.246$
	$z_q = 10$	$R_t = 0.3007$
		$R_q = 0.4625$

Test	Input	Output
10	$u = 9$	$u_* = 0.3495$
	$T = 12$	$\theta_* = -6.613 \times 10^{-2}$
	$q = 0.007$	$q_* = -1.059 \times 10^{-4}$
	$T_s = 14$	$z_0 = 2.556 \times 10^{-4}$
	$z_u = 10$	$z/L = -9.575 \times 10^{-2}$
	$z_t = 30$	$R_r = 5.956$
	$z_q = 10$	$R_t = 0.2642$
		$R_q = 0.4142$

Test	Input	Output
11	$u = 9$	$u_* = 0.3489$
	$T = 12$	$\theta_* = -6.919 \times 10^{-2}$
	$q = 0.007$	$q_* = -1.016 \times 10^{-4}$
	$T_s = 14$	$z_0 = 2.501 \times 10^{-4}$
	$z_u = 10$	$z/L = -9.777 \times 10^{-2}$
	$z_t = 10$	$R_t = 5.818$
	$z_q = 30$	$R_t = 0.2706$
		$R_q = 0.4227$

Test	Input	Output
12	$u = 6$	$u_* = 0.2359$
	$T = 7$	$\theta_* = -3.787 \times 10^{-2}$
	$q = 0.004$	$q_* = -1.016 \times 10^{-4}$
	$T_s = 8$	$z_0 = 1.536 \times 10^{-4}$
	$z_u = 5$	$z/L = -7.009 \times 10^{-2}$
	$z_t = 15$	$R_r = 2.416$
	$z_q = 25$	$R_t = 0.6049$
		$R_q = 0.8743$

ACKNOWLEDGMENTS

The authors wish to thank Alwilter T. Wilson of NRL's Software Systems Support Branch for her translation of the program. The publication of this report was made possible by a grant from the General Science and Technology Directorate of the Naval Research Laboratory.

REFERENCES

Liu, W.T., K.B. Katsaros, and J.A. Businger, 1979: Bulk Parameterizations of Air-Sea Exchanges of Heat and Water Vapor Including the Molecular Constraints at the Interface, *J. Atm. Sci.* **36**, 1722-1735.

APPENDIX

The specific humidity (q) in kg/kg is calculated by

$$q = \frac{0.622e}{P - (0.378e)}$$

where P is the barometric pressure and e is the water vapor pressure in pascals. If the humidity is measured by the wet and dry bulb temperatures (T_{wb}, T) in °C then

$$e = e_s - 6.6 \times 10^{-4}\left[1 + (1.15 \times 10^{-3} T_{wb})\right] P(T - T_{wb})$$

where e_s is the saturated vapor pressure in pascals such that

$$e_s = Pa_1^{b_3} \times 10^{a_2 b_4 + a_4 b_5 + a_5 b_6}$$

where

$$a_1 = \frac{373.16}{T + 273.16},$$

$$a_2 = a_1 - 1,$$

$$a_3 = 1 - \frac{1}{a_1},$$

$$a_4 = (10^{a_2 b_1}) - 1,$$

$$a_5 = (10^{a_3 b_2}) - 1,$$

and the Goff-Gratch humidity formulation constants are

$b_1 = -3.49149,$ $b_4 = -7.90298,$
$b_2 = 11.344,$ $b_5 = 8.1328 \times 10^{-3},$
$b_3 = 5.02808,$ $b_6 = -1.3816 \times 10^{-7}.$

If the humidity is measured by the dew point temperature (T_{dp}) in °C then

$$e = Pa_1^{b_3} \times 10^{a_2 b_4 + a_4 b_5 + a_5 b_6}$$

where

$$a_1 = \frac{373.16}{T_{dp} + 273.16}$$

and a_2 through a_5 are calculated in the same manner as for the saturated vapor pressure. If the relative humidity (RH) in % is measured then

$$e = \frac{e_s RH}{100}$$

where the saturated vapor pressure (e_s) is calculated from the air temperature as shown above.

GPO 907-178